

MATH 2280-1
MAPLE PROJECT : EARTHQUAKES
March 6, 2006

This handout might help for the Earthquake project in section 5.3 of the text, on pages 326-328. You are mostly on your own for this project, but here is a small example of a spring system worked out on Maple, so that you can get an idea about useful commands to use. This is the example we did by hand on Friday....

This is example 1 on page 317 of Edwards-Penney. Initially it is an unforced system with two masses and two springs, as you can see from the description on page 317. We can write the system as $\mathbf{M}\mathbf{x}'' = \mathbf{K}\mathbf{x}$, where \mathbf{M} is the “mass matrix”, \mathbf{K} is the “spring matrix”, and \mathbf{x} is the displacement vector. Following the book’s notation, we enter

```
[ > with(linalg):with(plots):with(DEtools): #tools for project
> M:=matrix([[2,0],[0,1]]);
  K:=matrix([[-150,50],[50,-50]]);
  A:=evalm(inverse(M)*K);
```

$$M := \begin{bmatrix} 2 & 0 \\ 0 & 1 \end{bmatrix}$$

$$K := \begin{bmatrix} -150 & 50 \\ 50 & -50 \end{bmatrix}$$

$$A := \begin{bmatrix} -75 & 25 \\ 50 & -50 \end{bmatrix}$$

Then the system can also be written as $\mathbf{x}'' = \mathbf{A}\mathbf{x}$; as we discussed on Friday, the eigenvectors of \mathbf{A} determine fundamental modes, and the corresponding negative eigenvalues are the (opposites) of the squares of the corresponding angular frequencies:

```
[ > eigenvectors(A);
      [-100, 1, {[ -1, 1]}], [-25, 1, {[ 1, 2]}]
```

Therefore, the natural frequencies of this system are the 10 and 5, and the two fundamental modes correspond to the masses moving in opposite directions (with equal amplitudes and angular frequency 10) and in parallel directions (with amplitude ratio of two and angular frequency 5).

Now, let’s consider the forced system with force vector equal to $\cos(\omega t)[0,50]$, i.e. the second mass is being forced periodically. In other words, the system $\mathbf{M}\mathbf{x}'' = \mathbf{K}\mathbf{x} + \mathbf{F}$, where $\mathbf{F} = \cos(\omega t)[0,50]$; this is Example 3 on page 323, and we also discussed this on Friday. We follow the method described on that page to find a particular solution to the forced oscillation problem, of the form given by equation (31). Here is the Maple version of the details summarized in the text:

```
[ > F0:=evalm(inverse(M)*vector([0,50]));
      #The F0 in the normalized equation (32), page 323
  Iden:=array(1..2,1..2,identity);
      #the 2 by 2 identity matrix
  Aleft:=omega->evalm(A + omega^2*Iden);
      #the matrix function multiplying
```

```

# c on the left side of (32)
c:=omega->evalm(-inverse(Aleft(omega))*F0);
# the solution vector c(omega) to (32),
# obtained by multiplying both sides of equation
# (32) on the left, by the inverse to Aleft

```

$$F0 := [0, 50]$$

$$Iden := \text{array}(\text{identity}, 1..2, 1..2, [])$$

$$Aleft := \omega \rightarrow \text{evalm}(A + \omega^2 Iden)$$

$$c := \omega \rightarrow \text{evalm}(-\&*(\text{inverse}(Aleft(\omega)), F0))$$

```

> c(omega); #see equation (35) page 323

```

$$\left[\frac{1250}{2500 - 125\omega^2 + \omega^4}, -\frac{50(-75 + \omega^2)}{2500 - 125\omega^2 + \omega^4} \right]$$

The vector $c(\omega)$ above, times the oscillation $\cos(\omega t)$, is a particular solution to the forced oscillation problem we are considering. If we assume that our actual problem has a small amount of damping, then we expect that this particular solution is very close to the steady periodic solution to the damped problem. See the discussion on page 324. We can study resonance phenomena for these slightly damped problems by plotting the maximum amplitude of the steady state solutions to the undamped problems. That would be the maximum absolute value of c_1 and c_2 above. Use the Maple command “norm” to measure this maximum amplitude:

```

> norm(c(omega));

```

$$\max\left(\left|\frac{1250}{2500 - 125\omega^2 + \omega^4}\right|, 50\left|\frac{-75 + \omega^2}{2500 - 125\omega^2 + \omega^4}\right|\right)$$

Another way to measure the size of $c(\omega)$ is to take its Euclidean magnitude, which is the command

```

> norm(c(omega), 2);

```


$$50\sqrt{\frac{625}{|2500 - 125\omega^2 + \omega^4|^2} + \left|\frac{-75 + \omega^2}{2500 - 125\omega^2 + \omega^4}\right|^2}$$

(You will use the first command in the Earthquake project, which perhaps makes the most sense since it will be measuring the maximum amplitude that any floor oscillates.) The following picture illustrates that the maximum amplitude of the particular solution blows up when ω is near the two natural angular frequencies. Thus, in the slightly damped problem, one would experience practical resonance in the steady periodic solution.

```

> plot(norm(c(omega)), omega=0..15, amplitude=0..15,
      numpoints=200, color='black');


```



This is qualitatively the picture on page 323, figure 5.3.10, although they plotted the Euclidean magnitude of $c(\omega)$ rather than the maximum amplitude. Notice how we get Maple to label the axes as desired

We can get a plot of resonance as a function of period by recalling that $2\pi/T = \omega$:

```
> plot(norm(c(2*Pi/period)), period=0.1..3, amplitude=0..15,
      numpoints=200, color='black');
```



COMMENTS FOR THE EARTHQUAKE PROJECT:

(1) Students are often confused by the forcing term in equation (2) of page 327, namely

```
> E*(omega)^2*cos(omega*t)*b;
```

$$E \omega^2 \cos(\omega t) b$$

where b is the transpose of $[1,1,1,1,1,1,1]$. They ask, “how can the earthquake be forcing all seven stories, it seems like it’s just shaking the bottom one.” Well, the students are correct, but so is Edwards-Penney. The authors talk about an “opposite inertial force” being the reason for this forcing term and there’s a detailed discussion of this on page 4 of Friday’s notes. here’s a brief summary. Think of the ground as the zeroth story. In the rest frame it is shaking with oscillation $E \cos(\omega t)$. And so its acceleration is its second time derivative, namely $-E \omega^2 \cos(\omega t)$. If you write down the inhomogeneous system of EIGHT second order DE’s for the accelerations of stories zero thru seven, the forcing (well, accelerating) term is $-E \omega^2 \cos(\omega t) [1,0,0,0,0,0,0]$, as you would expect. Call the solution 8-vector to this system $y(t)$, then see what the shaking looks like to someone on the ground by

letting $\mathbf{x}(t) = \mathbf{y}(t) - E \cos(\omega t) [1, 1, 1, 1, 1, 1, 1]$. Then the zeroth story component of $\mathbf{x}(t)$ will be identically zero, and the other seven components will satisfy equation (2) on 327, exactly as the authors claim.

Very important note:

(2) For large matrices the `eigenvect` command won't work well unless you enter at least one decimal number; if all entries are rational numbers (expressed without decimal points), Maple tries to find the eigenvalues and eigenvectors algebraically and exactly, instead of numerically, and often fails. **Make sure at least one of your matrix entries has a decimal point in it.**